



**MCP-40**

COLOUR PRINTER

**Operation  
Manual**

## **Introduction**

Congratulations for selecting this high-quality, low-cost colour printer/plotter! We know this Printer is the ideal output device for home or business use.

Its special features include:

- A set of simple instructions that allow you to change colours, backspace the pens, reverse feed the paper, and more!
- Four colour printout (black, blue, green, and red ).
- Standard roll paper (4.48" wide by 56meter long).
- Selectable 40-character or 80-character per line printing.

## Contents

1 General Description .....	1
2 Setting Up .....	2
3 Background Information .....	6
4 Using the printer/plotter .....	10
Control Codes .....	11
Graphic Commands .....	12
5 Specifications .....	24
Appendix A / Sample Programs .....	26
Appendix B / Graphic Commands Summary .....	31
Appendix C / ASCII Character Set .....	32

## 1 General Description




Figure 1. Front View





Figure 2. Rear View


## 2 Setting Up

### Power Cable Connections.




Connect the power cable from the AC Power input of the printer (on rear panel) to a wall outlet.

### Connecting the printer/plotter to your Computer



1. Connect one end of the cable to the printer/plotter.
2. Connect the other end of the cable to the Computer.




## Paper Loading

When loading paper into the printer/plotter, follow these steps:

1. With scissors, cut the end of the paper square.
2. Insert the end of the paper into the slot in the bottom of the Printer.
3. Insert the shaft into the roll and place the roll of paper in the paper compartment.
4. Turn the power on.
5. Raise the top cover so printing will not be done on the platen.
6. Press the LINE FEED button to feed the paper in through the slot and around the platen.

## Pen Installation

Be very careful when touching the Pen Holder. The Holder is delicate and can be easily damaged. Never rotate or move the Pen Holder manually. Use the printer controls to move the Pen Holder.



Pen Holder Mechanism

## Pen Installation/Removal

Important Note: You must always use four pens.  
Using the Printer with only three (or fewer) pens may cause incorrect colour changes when you are changing colour.

You can install the Pens in any order you need. However, the Printer recognizes four specific pen Positions--Pen #1, Pen #2, Pen #3, and Pen #4. We suggest you use the sequence in the Table below when installing pens. When referring to Pen Colours and Pen Positions in this manual, we'll be using this sequence

Pen Positions and Pen Colours	
Pen #	Colour
1	Black
2	Blue
3	Green
4	Red

Table

There are two ways to identify a particular Pen Position :

- The Pen Guide Wheel is colour-coded. Look carefully near the center of the Wheel and match the colour of the Pen with the colour on the Wheel.
- Turn the Printer OFF, then back ON. On power-up, Pen # 1 (colour-coded Black) is always up. To get to Pen # 2, press COLOUR SELECT; to get to Pen # 3, press COLOUR SELECT again.

We recommend that if you know that you won't be using the Printer for a while (e.g., more than a day or two), that you remove the Pens from the Printer and "cap" them so the ink won't dry out. Rub the Pen Point back and forth on a piece of paper to get the ink flowing before you install it in the Printer.

Before you install, remove, or replace any Pens, take a look at (and become familiar with) what's under the cover.

Raise the top cover and look at the right side printing area. You'll see a wire about 1/2" long pointing to the left. Above it, and to the right, there will be a small white lever--the Pen Ejection Lever. Gently press down on this lever and the wire will go up. When a Pen is in the Holder, this will "pop" the Pen out.

Note: Do not move or rotate the Pen Holder manually.

To install, remove, or change Pens:

1. Raise the Top Cover.
2. Press the COLOUR SELECT Switch to rotate to the Pen Position # you need.
3. Press the PEN CHANGE Switch

The Pen Holder will move to the far right. Note that the Pen Ejection Lever wire will be under the Pen.

4. If you're removing or changing a Pen:
  - Gently hold the Pen down with one finger.
  - Gently press down on the Pen Ejection Lever.

If the Pen "pops" into the Printer, hold the Paper Roll (if installed), turn the Printer upside-down, and gently shake the Printer until the Pen falls out. Do not operate the Printer with a Pen loose in the printing mechanism. Damage to the Printer may result.

- Remove the pen.

To install a new Pen, insert the "ink" tip of Pen into the Pen Wheel Guide hole. (Match the colour on the pen with the colour on the Wheel.) Gently "snap" the other end of the Pen into place.

5. To rotate the Pen Holder to the next Pen Position (#2), press the COLOUR SELECT Switch.

The Pen Holder will move to the left, rotate to Pen Position # 2, and return to the right most position.

To select Pen Position # 3, repeat the above procedure.

6. Close the Top Cover.
7. Press the LINE FEED Switch and the Pen Holder will return to the left margin and will be ready to begin printing.

### 3 Background Information

The BASIC command to send information to the printer is LPRINT.

```
LPRINT "V" (RETURN)
```

Before printing, the Printer checks to see if the character sent is an instruction (on how to print) instead of data (to be printed). An instruction, for example, might tell the Printer to backspace or change to a different Pen.

Consequently, some ASCII codes were created as instructions to control printers (and therefore are called "Control Codes").

### How do we send instructions to the Printer?

To send instructions to the Printer, use "control codes." To do this, use the BASIC function CHR\$( ). For example, to tell the Printer to backspace (ASCII Code 08), use the command:

```
LPRINT CHR$(08) (RETURN)
```

In a program (or in the "direct mode") and the Printer will backspace.

Any ASCII code (control codes as well as data) can be sent to the Printer this way. Appendix C lists the ASCII codes the Printer recognizes. For instance, Appendix C will tell you that the ASCII code for the letter Z is 90 (decimal). If you use the command:

```
LPRINT CHR$(90) (RETURN)
```

Printer will print the letter Z.

### Text and Graphic Modes

The Printer has two modes of operation:

- Text Mode for word processing, note writing, program listings, and program output.
- Graphic Mode for generating graphs, pictures, and other visual creations.

**Text Mode.** On initial power-up, the printer will be in Text Mode. When in Graphic Mode, to return to Text Mode, type:

```
LPRINT CHR$(17) (RETURN)
```

**Graphic Mode.** In this mode, you can generate graphs, pictures and other visual creations. To put the Printer in Graphic Mode, send a Control Code 18 (decimal) to the Printer using LPRINT. For instance. :


```
LPRINT CHR$(18) (RETURN)
```

In Graphic Mode, you can move the Pen to any point on the paper and draw a line to any other point.

Think of the Pen as being on a Cartesian coordinate plane, with the X-axis running left and right (horizontal) and the Y-axis going up and down (vertical).

Positive is up and to the right, negative is to the left and down the paper. See Figure below.

The origin may be set anywhere on the paper. When first entering the Graphic Mode, the origin is at the left margin and under the Pen.



**Coordinate (X-Y) Axis**

Generally speaking, you can tell the Pen to do two things:

- MOVE (with the Pen up)
- DRAW (with the Pen down).

The Pen can move (or draw) two different ways:

- RELATIVE Movement (from the current Pen position to a point relative to the pen location.)
- ABSOLUTE Movement (from the current Pen position to a point relative to the Origin.)

There are times when a specific method will be more convenient to use.

## How does the Pen move?

In both horizontal and vertical directions, Pen movement is measured in "steps." Each step is 0.2 mm long (about the size of a period), and there are 480 steps across the paper.

## Care and Maintenance

The Printer is a very reliable unit that should give years of satisfactory service. However, there are some definite Do's and Don't's:

- Do not manipulate the Pen Holder manually. This will very easily cause damage to the Printer. Be especially careful when inserting and removing the pens.
- The Pen Holder is self-propelled. Do not move it manually. To move it to the right, press the PEN CHANGE Switch. To move it to the left, press LINE FEED. To rotate the Pen Holder press COLOUR SELECT.
- Do not operate the Printer without Pens. Leave a dry or broken pen in the Holder until you can replace it.
- Cap the pens and return them to their case when the Printer is not in use.
- Do not attempt to use Pens or paper not designed for this Printer.

The Printer contains a self-test routine. To see all the characters, turn the Power Switch on while holding down the LINE FEED Switch. You will see:

```

! " # $ % & ' ( ) * + , - . / 0 1 2 3 4 5 6 7 8 9 : ; < = > ? @ A B C D E F G
H I J K L M N O P Q R S T U V W X Y Z [ \ ] ^ _ ` a b c d e f g h i j k l m n o
p q r s t u v w x y z { | } ~ ☐

```

### Self Test

## 4 Using the Printer/Plotter

### Automatic Start-up Sequence

Once the Printer is properly connected, it will execute a short, built-in routine that resets the Pen Holder and draws four small boxes in each available pen color on power-up.

This allows you to be sure that each Pen has enough ink and is drawing correctly before you begin using the Printer.

If the Top Cover is open on power-up, the Printer will execute the routine but the Pens will not touch the paper. This will prevent the Pens from drawing on the Platen when paper is not being loaded.



### Automatic Start-Up Sequence

After the four boxes are drawn, the Pen will return to the left margin. The printer will then be in Text Mode and ready for operation.

## Manual Operation

You can change pen colour with the COLOUR SELECT key and advance the paper using the LINE FEED key.

## Program Control

The printer can be controlled entirely from within a BASIC program. Note that you can use Control Codes (CHR\$( )) to send instructions to the Printer as well as data to be printed. In this section, we'll describe the Control Codes in sequence starting from CHR\$(08) through CHR\$(29). Then we'll describe the Graphic Commands listed alphabetically.

With each code or command, we've included a short example. The examples serve two purposes. First, they show how to use the command being discussed. Second, they show how to use combinations of commands to draw with the Printer. Once you see how

simple it is to program the Printer, you can write your own graphic programs. (For more involved programs, see Appendix A.)

## Control Codes

### CHR\$(8) Backspace (Text Mode)

This command backspaces the Pen one character at a time. CHR\$(8) is very useful when you need to underline.

#### Example

```
10 LPRINT "A";  
20 LPRINT CHR$(8);  
30 LPRINT CHR$(95)
```

This program will print the letter A, then backspace (line 20) and print an underline.

### CHR\$(11) Reverse Line Feed (Text Mode)

CHR\$(11) moves the paper backwards (reverse) one line at a time. This is often used for superscripts.

#### Example

```
10 REM SUPERSCRIPT DEMONSTRATION  
20 LPRINT "2";  
30 LPRINT CHR$(11);  
40 LPRINT "2";  
50 LPRINT CHR$(10) : REM LINEFEED  
60 END
```

### CHRS(18) Select Graphic Mode

When the Printer is in Text Mode, this command will put it into Graphic Mode.



Example

```
10 LPRINT CHR$(18)
```

## CHR\$ (17)

### Select Text Mode

This command is used when text material is to be printed. There are other commands that allow you to print words in Graphic Mode, but CHR\$ (17) is the easiest way to change to Text Mode and print text between graphic material.

Example

```
20 LPRINT CHR$(17)
```

## CHR\$ (29)

### Rotate Pen Holder (Text Mode)

This command advances the Pen Holder one colour. CHR\$ (29) is used to change colour from Text Mode and you must keep track of what colour is next. If you want a colour beyond the next Position, you may use the command more than once.

Example

```
10 LPRINT CHR$(29)
```

## Graphic Commands

The following commands can be used in Graphic Mode only.

## A

### Return to Text Mode (Reset)

```
A
```

This command moves the Pen Holder to the left margin (without drawing a line and without vertical movement) and returns the Printer to text Mode. In this case, the Origin is also re-defined (reset) as the left margin.

Example

```
10 LPRINT CHR$(18)
20 LPRINT "A"
30 LPRINT "A"
```

This example will put the Printer into Graphic Mode (line 10). Line 20 will return the Printer to Text Mode. Line 30 will print the letter A.

## C

### Change Colour

```
C colour
```

colour is a numeric expression from 0 to 3. colour is optional; if omitted, 0 is used.

If you installed the Pens according to the suggested sequence earlier, then 0 = Black, 1 = Red, 2 = Green, and 3 = Blue.

If not, then the Pen you installed in Pen Position # 1 will determine the colour when you send the |Colour command to the printer.

Example

```
10 LPRINT "C1"
```

The following sample program illustrates how the Command works. It also illustrates CHR\$(18) which is the Graphics Select command.

```
10 REM COLOUR SELECT
20 LPRINT
30 LPRINT CHR$(18):LPRINT"C0"
40 LPRINT CHR$(17)::LPRINT"I LOVE YOU"
50 LPRINT CHR$(18):LPRINT"C1"
60 LPRINT CHR$(17)::LPRINT"JE T'AIME"
70 LPRINT CHR$(18):LPRINT"C2"
80 LPRINT CHR$(17)::LPRINT"YO TE AMO"
90 LPRINT CHR$(18):LPRINT"C3"
100 LPRINT CHR$(17)::LPRINT"ICH LIEBE DICH"
110 LPRINT CHR$(18):LPRINT"C0"
120 LPRINT "A"
130 END
```

Running this program will result in:

```
I LOVE YOU
JE T'AIME
YO TE AMO
ICH LIEBE DICH
```

## D Draw (Absolute)

**D** *destination...*

*destination* specifies the endpoint of the point you wish and is a X-Y coordinate. (The startpoint of the line is the current Pen position.) *destination* may be repeated to draw more than one line.

D draws a line from the current pen position to a destination point. The destination point (e.g., the point where the line ends) is in the form x,y where X and Y are a coordinate pair with respect to the Origin.

If more than one pair of coordinates are specified then the line will be continued to the second point, then to the third point, etc. X and Y are values between -999 and 999.

**Example**

```
10 REM DRAWING EXAMPLE
20 LPRINT CHR$(18) : REM TURN ON GRAPHICS
30 LPRINT "D0,100,100,100,100,0,0,0"
40 LPRINT "A"
50 END
```

The example will draw a box. Remember that the paper is 480 steps across and as deep as you wish (up to 999 steps). Since you didn't specify a different Origin, the Pen started at 0,0. The corners of the box are at 0,0 0,100 100,100 and 100,0.

## H Return to Origin (Home)

**H**

The H command will move the Pen to the Origin without drawing a line.

**Example**

```
10 LPRINT "H"
```

## I Set Origin (Initialize)

**I**

This command resets the Origin to the current location of the Pen point.

### Example

```

10 REM RELOCATE ORIGIN
20 LPRINT CHR$(18)
30 LPRINT "D240, 0"
40 LPRINT "I"
50 END

```

This short routine draws a line to the center of the paper, then defines the center of the paper as the Origin.

## J

### Draw (Relative)

**J** *destination...*  
*destination* specifies the endpoint of the point you wish to draw to and is a X-Y coordinate. (The startpoint of the line is the current Pen position.) *destination* may be repeated to draw more than one line.

The J command draws a line from the current Pen position to an end point. The end point is determined by measuring up X units and right Y units. (or left and down if X and Y are negative).

Once the Pen is at this new position, the line can be continued to another point by supplying another pair of X-Y values. The new point is measured from the previous point, and not from the original pen position. X and Y must be in the range -999 to 999.

### Example

```

10 REM RELETIVE DRAWING
20 LPRINT CHR$(18)
30 LPRINT "J0, 100, 100, 0, 0, -100, -100, 0"
40 END

```

This draws the same box as the sample for the D command, but specifies the corners differently. Read line 30 like this:

"From where you are, draw a line to the point that is 0 steps to the right (in the X-direction) and 100 steps up (in the Y-direction). Then from that point, draw a line that is 100 steps to the right and 0 steps up, then a line from that is 0 steps to the right and 100 steps down. Finally draw a line that is 100 steps to the left and 0 steps up."

## M Move (Absolute)

### *Mx,y*

*x* specifies a position on the X-axis and is a numeric expression between -999 and 999.  
*y* specifies a position on the Y-axis and is a numeric expression between -999 and 999.

The M command moves the Pen from its present location to the point specified by x,y without drawing a line. X and Y must be in the range -999 to 999.

### Example

```
10 LPRINT "M100, -100"
```


## L Line Type

### *L type*

*type* specifies the "type" of line you wish to draw and is a numeric expression from 0 to 15. *type* is optional, if omitted, 0 is used. 0 is a solid line. 1 through 15 generate different types of dash lines, the greater the number, the farther apart the dashes are.

The L command lets you specify different line types. You can specify a solid line or 15 different types of dashed (or dotted) lines. The following table illustrates the line types available:

Line Types	
Line Specified	Line Drawn
0	—————
1	.....
2	.....
3	.....
4	.....



**Example**

```

10 REM LINE SAMPLE PROGRAM
20 LPRINT CHR$(18)
30 LPRINT "L3"
40 LPRINT "J430,0"
50 LPRINT "A"
60 END

```

Line 30 sets the line type to 3. The program will draw a dotted line across the paper. Add an instruction to change the colour if you wish.

## P Print Text Characters

**P characters**

*characters* is either an alpha-character (A-Z) or a numeric-character.

The P command lets you print either alpha- or numeric-characters while in Graphic Mode without first returning to Text Mode. Characters may any string of letters or numerals.

After the command is executed, the Printer will still be in Graphic Mode.

**Example**

```
10 LPRINT "PGRAPHIC PRINTER"
```

Will print the string GRAPHIC PRINTER while the Graphic Printer is in Graphic Mode.

## S Character Size

**S size**

*size* specifies the size of the printed characters and is a numeric expression between 0-63. *size* is optional; if omitted, 0 is used.

The S command lets you specify the size of the character drawn with the P command. *size* may be a numeric expression between 0-63; 0 draws the smallest characters size (80 characters per line), 63 the largest (1 character per line).

Use this formula to determine the size of printed character in relation to the numeric value you specify:

$$cpl = 80 / (size + 1)$$

where cpl is the number of characters per line and size is the numeric value you specify with the S command.

## Q Rotate Print Direction

**Q direction**

*direction* specifies which direction you wish printing to be and is a numeric expression from 0 to 3. *direction* is optional; if omitted, 0 is used.

On power-up (in Text Mode), Q0 is used.

However, when you enter Graphic Mode, you can specify one of four directions for character printing. See Table below.

Printing Direction	
Direction specified	Print Direction
0	Left-to-Right
1	Top-to-Bottom
2	Right-to-Left (Upside down)
3	Bottom-to-Top

For instance, using the word JON as an example:

• Q0 (left-to-right) would produce JON.

• Q1 (top-to-bottom) would print  
J  
O  
N

• Q2 (right-left) would produce NOJ

• Q3 (bottom-to-top) would print  
N  
O  
J

#### Example

```

10 REM WRITING SIDWAYS
20 LPRINT CHR$(18)
22 LPRINT "M50,0" : REM ALLOW LEFT MARGIN
30 INPUT"TYPE YOUR NAME " ;N$
40 LPRINT "P";N$ : REM WRITE YOUR NAME
50 LPRINT "Q1" : REM CHANGE DIRECTION
60 LPRINT "P";N$ : REM WRITE YOUR NAME
70 LPRINT "Q2" : REM NOW UPSIDE DOWN
80 LPRINT "P";N$ : REM WRITE YOUR NAME
90 LPRINT "Q3" : REM CHANGE DIRECTION
100 LPRINT "P";N$ : REM WRITE YOUR NAME
110 LPRINT "Q0" : REM BACK TO RIGHTSIDE UP
120 LPRINT "A" : REM BACK TO TEXT MODE
130 END

```

## R Move (Relative)

**R** *x,y*

*x* specifies a point on the X-axis and is a numeric expression between -999 and 999.

*y* specifies a point on the Y-axis and is a numeric expression between -999 and 999.

This command moves the from the current location to the point that is *x* steps away to the right (or left if *x* is negative) and *y* steps up (down if *y* is negative).

#### Example

```

10 REM SAMPLE OF RELATIVE MOVE
20 LPRINT CHR$(18) : REM TURN ON GRAPHICS MODE
30 LPRINT "R100,0" : REM MOVE TO RIGHT 100 STEPS
40 LPRINT "R0,-100" : REM NOW MOVE 100 STEPS DOWN
50 LPRINT "J-100,100" : REM NOW DRAW A LINE BACK TO THE
BEGINNING
60 LPRINT "A" : REM AND QUIT
70 END

```

1. Line 30 moves the point 100 steps to the right and 0 steps up.
2. Line 40 moves the point 100 steps down and 0 steps to the right.
3. Line 50 draws relative back to the origin, undoing the moves of lines 30 and 40.

## X Draw X-Y Axis

**X** *axis, step, interval*

*axis* specifies axis to be drawn and is a numeric expression of either 1 (X-axis) or 0 (Y-axis).

*step* specifies the graduation(distance) between measurement marks on the axis and is a numeric expression between -999 and 999.

*interval* specifies the number of times that *step* is to be repeated and is a numeric expression between 1 and 255.

The X command lets you draw Coordinate (X-Y) Axis, divide the axis into specified units of measurements (graduations), and designate how many units of measurements are to be drawn on the axis.

If axis is 0, a vertical (Y) axis will be drawn. If axis is 1, a horizontal (X) axis will be drawn.

The axis can also be divided into segments. Since each step is 0.2mm, the exact distance between segment marks can be easily set. You may have 1 to 255 intervals, each one 1 to 999 steps long. If interval is negative, the axis will be drawn from the left, or down depending on the choice for axis.

#### Example

```
10 LPRINT "X0,6,20"
```


will draw a vertical axis up from the present location.  
There will be a small dash (grad) every 6 steps (20 dashes altogether).

```
10 LPRINT "X1,-10,16"
```

This command will produce a horizontal axis left of the current Pen location, with 16 sections each 10 steps apart.

```
10 REM X-Y AXIS TO BE DRAWN
20 LPRINT: LPRINT CHR$(10) :REM GO INTO GRAPHIC MODE
30 LPRINT "M240,-240" :REM MOVE TO SPECIFIED POINT
40 LPRINT "I" :REM SET NEW ORIGIN
50 LPRINT "X1,40,6" :REM HALF OF X-AXIS
60 LPRINT "H" :REM MOVE PEN TO ORIGIN
70 LPRINT "X1,-40,6" :REM OTHER HALF OF X-AXIS
80 LPRINT "H" :REM MOVE PEN TO ORIGIN
90 LPRINT "X0,20,12" :REM HALF OF Y-AXIS
100 LPRINT "H" :REM MOVE PEN TO ORIGIN
110 LPRINT "X0,-20,12" :REM OTHER HALF OF Y-AXIS
120 LPRINT "A" :REM RESET, MOVE PEN TO LEFT
130 END
```

This program will produce a Coordinate Axis such as this:



## 5 Specifications

Printing/Plotting System	Ball Point Pen, 4 colour
Plotting Speed (Horizontal)	52 mm/sec(2.05ips)
(Vertical)	73 mm/sec(3.80ips)
Printing Speed	12 characters per second
Resolution	0.2 mm/step(0.00787 inch)
Effective Plotting Range	96 mm (3.804 inch) x axis Divided into 480 steps (No limit in y direction)
Characters per Line	80 or 40 (Text Mode) (Determined by Software in Graphics Mode)
<p>Characters per Line = <math>\text{INT} (480/(n+1) * 6)</math> for <math>0 \leq n \leq 15</math></p>	
Accuracy (repetition)	0.2 mm max
(Movement)	0.3 mm max
(Distance)	0.5% max (X-axis) 1 % (Y-axis)
Dimensions	276 mm wide (8.4 in) 174 mm deep (8.64 in) 68 mm high (3 in)
Weight	850g
Pen Life	250 meters (825 feet)
Parallel Interface	8-bit parallel.



Selectable Modes	
Self Test	Prints 96 ASCII character set in 4 colours
Text Mode	Parallel Printing
Graphics Mode	Image Plotting using the Various commands
Temperature Range	18.3 to 35°C (65 to 96F°)
Storage	-40 to 71°C (-40 to 160 F°)
Humidity Range	10% to 80% relative non- condensing
Power Supply	Switching Power Supply
	Input 100~120 VAC
	200~240 VAC
	Output 5VDC 3A

# Appendix A Sample programs

```


100 REM *** SAMPLE PROGRAM ***
110 REM *** CHARACTER SET ***
120 LPRINT CHR$(18);"S1"
130 LPRINT"CU";LPRINT CHR$(17)
140 LPRINT:LPRINT
150 LPRINT TAB(14)"CHARACTER SET"
160 LPRINTCHR$(18);"S2";LPRINT CHR$(17)
170 FOR I=32 TO 127
180 LPRINT CHR$(I);
190 NEXT I
200 LPRINT:LPRINT CHR$(18);"S1"
210 LPRINT"C1":LPRINT CHR$(17)
220 REM *** ROTATE 0 TO 3 ***
230 LPRINT:LPRINT
240 LPRINT TAB(13)"ROTATE 0 TO 3"
250 LPRINT CHR$(18);LPRINT"S7"
260 LPRINT"H"
270 LPRINT"M85,-55"
280 LPRINT"C3"
290 FOR I=0 TO 3
300 LPRINT"Q";I:LPRINT"ROTATE"
310 LPRINT"C";I
320 NEXT I
330 LPRINT
340 REM *** LINE TYPE ***
350 LPRINT"M35,-500":LPRINT"S1"
360 LPRINT"A":LPRINT TAB(15)"LINE TYPE"
370 LPRINT CHR$(18);LPRINT"C0"
380 FOR I=0 TO 9
390 LPRINT"L";I
400 LPRINT"J 400,0"
410 LPRINT"R -400,-15"
420 NEXT I
430 LPRINT"S1":LPRINT"C3":LPRINT CHR$(17)
440 REM *** PLOTTING ABILITY ***
450 LPRINT:LPRINT TAB(12)"PLOTTING ABILITY"
460 LPRINT CHR$(18);"L0"
470 LPRINT"M250,-180":LPRINT"1"
480 FOR I=0 TO 350 STEP 10
490 S=I/130*3.14159
500 X=SIN(S)*200.5:Y=COS(S)*200.5
510 X=INT(X):Y=INT(Y)
520 LPRINT"D";X;",";Y:LPRINT"H"
530 NEXT I
540 LPRINT "M0,-450"
550 LPRINT"1"
560 S=2*3.141592625#/11
570 FOR I=0 TO 10
580 A(I,1)=INT(SIN(I*S)*200.5)
590 A(I,2)=INT(COS(I*S)*200.5)
600 NEXT I
610 LPRINT"M";A(0,1);",";A(0,2)
620 C=2
630 FOR I=0 TO 4
640 K=0
650 C=C+1:IF C>3 THEN C=0
660 LPRINT"C";C
670 FOR J=0 TO 10
680 K=K+1+1
690 IF K>10 THEN K=K-11:GOTO 690
700 LPRINT "D";A(K,1);",";A(K,2)
710 NEXT J
720 NEXT I
730 LPRINT"D";A(K,1);",";A(K,2)
740 LPRINT"M0,-200":LPRINT"C0"
750 LPRINT CHR$(17)
760 END

```



CHARACTER SET  
! " # \$ % & ' ( ) \* + , - . / 0 1 2 3 4 5 6 7 8 9  
: ; < = > ? @ A B C D E F G H I J K L M N O P Q R S  
T U V W X Y Z [ \ ] ^ \_ ` a b c d e f g h i j k l m  
n o p q r s t u v w x y z { | } ~ ¢

ROTATE 0 TO 3  
ROTATE  
ROTATE  
ROTATE



```

1000 REM *****
1010 REM *** 4 Pen Colour Graphic Micro Plotter/Printer MCP-40 **
1020 REM *****
1030 REM
1040 DIM A$(5),DDATA(5),C(5),ST(5),D(5)
1050 REM *** GRAPH DEMO ***
1060 LPRINT CHR$(18)
1070 LPRINT"C3"
1080 LPRINT"S2":LPRINT"A"
1090 LPRINT"4 PEN Colour Graphic"
1100 LPRINT CHR$(11);CHR$(18)
1110 LPRINT"C2":LPRINT"S2"
1120 LPRINT"A":LPRINT" Micro Plotter/Printer"
1130 LPRINT CHR$(11);CHR$(18)
1140 LPRINT"C1"
1150 LPRINT"S5"
1160 LPRINT"A"
1170 LPRINT" MCP-40"
1180 REM *** SIN & COS GRAPH ***
1190 REM
1200 LPRINT CHR$(18);LPRINT"H"
1210 LPRINT"S2"
1220 LPRINT"C0"
1230 LPRINT"M44,0"
1240 LPRINT"P*** SIN & COS GRAPH ***"
1250 LPRINT"M0,-45"
1260 LPRINT"C2"
1270 LPRINT"PSIN"
1280 LPRINT"R5,10"
1290 LPRINT"J30,0,0,3,-30,0,0,-3"
1300 LPRINT"M0,-70"
1310 LPRINT"C3"
1320 LPRINT"PCOS"
1330 LPRINT"R5,10"
1340 LPRINT"J30,0,0,3,-30,0,0,-3"
1350 REM *** PLOT AXIS ***
1360 LPRINT "C0"
1370 LPRINT"M240,-200"
1380 LPRINT"X1,60,4"
1390 LPRINT"M240,-200"
1400 LPRINT"X1,-60,4"
1410 LPRINT"L5"
1420 LPRINT"M120,-100":LPRINT"D120,-299"
1430 LPRINT"M240,-100":LPRINT"D240,-299"
1440 LPRINT"M360,-100":LPRINT"D360,-299"
1450 LPRINT"M0,-320"
1460 LPRINT"P0 PI 2PI "
1470 LPRINT"M0,-340"
1480 LPRINT"P 1/4PI 3/4PI "
1490 REM *** SIN GRAPH ***
1500 LPRINT"C2":LPRINT"L0"
1510 FOR I=0 TO 3.142*2 STEP .02
1520 Y=200-SIN(I)*99
1530 X=I/(3.142*2)*480
1540 IF I=0 THEN LPRINT"M";X;",";"-Y
1550 LPRINT"D";X;",";"-Y
1560 NEXT
1570 REM *** COS GRAPH ***
1580 LPRINT"C3"
1590 FOR I=0 TO 3.142*2 STEP .02
1600 X=I/(3.142*2)*480
1610 Y=200-COS(I)*99
1620 IF I=0 THEN LPRINT"M";X;",";"-Y
1630 LPRINT"D";X;",";"-Y
1640 NEXT
1650 REM *** BAR GRAPH ***
1660 REM
1670 LPRINT CHR$(18);LPRINT"M0,-400":LPRINT"A":LPRINT CHR$(18);LPRINT"1"
1680 FOR I=1 TO 5
1690 READ A$(I),DDATA(I),C(I),ST(I)
1700 NEXT
1710 S=0
1720 FOR I=1 TO 5
1730 S=S+DDATA(I)
1740 NEXT
1750 GOSUB1970

```



```

1760 FOR I=1 TO 5
1770 H=DDATA(I)/SX100
1780 IF ST(I)=5 THEN NN=2
1790 IF I<=4 THEN NN=1
1800 GOSUB1830
1810 NEXT
1820 GOTO 2210
1830 REM **** GRAPH LINE ****
1840 REM
1850 LPRINT"C";C(I):LPRINT"M";I*50;",";-199
1860 LPRINT"D";I*50;",";HX2-200;",";I*50+40;",";HX2-200;",";I*50+40;",";-199
1870 ON NN GOSUB 1890,1930
1880 RETURN
1890 FOR J=199 TO 200-HX2 STEP -ST(I)
1900 LPRINT"M";I*50;",";-J:LPRINT"D";I*50+40;",";-J
1910 NEXT
1920 RETURN
1930 FOR J=0 TO 40 STEP ST(I)
1940 LPRINT"M";I*50+J;",";-200+HX2:LPRINT"D";I*50+J;",";-199
1950 NEXT
1960 RETURN
1970 LPRINT"C0"
1980 LPRINT"M25,-199":LPRINT"X0,20,10"
1990 LPRINT"M25,-199":LPRINT"D25,-199,350,-199"
2000 LPRINT"LB"
2010 LPRINT"M25;",";-100:LPRINT"D25;",";-100;",";350;",";-100
2020 LPRINT"L0"
2030 LPRINT"M0,0"
2040 LPRINT"S1"
2050 LPRINT"P **** BAR GRAPH ****"
2060 LPRINT"M0,-199"
2070 LPRINT"S0"
2080 LPRINT"P 0%"
2090 LPRINT"M0,-99"
2100 LPRINT"P 50%"
2110 LPRINT"M0,0"
2120 LPRINT"P100%"
2130 RETURN
2140 DATA A INC,42510,1,3
2150 DATA B INC,25350,2,3
2160 DATA C INC,20124,3,3
2170 DATA D INC,11200,4,3
2180 DATA E INC,04651,1,5
2190 REM
2200 REM *** PI GRAPH ***
2210 RESTORE
2220 LPRINT CHR$(18):LPRINT"C0"
2230 LPRINT "M0,-300":LPRINT"S2":LPRINT"A":LPRINT " **** PI GRAPH ****"
2240 LPRINT CHR$(18):LPRINT"I":LPRINT"S1"
2250 LPRINT"M0,-100,100,-100"
2260 LPRINT"I"
2270 FOR I=1 TO 5
2280 READ A*(I),DDATA(I),C(I),ST(I)
2290 NEXT
2300 S=0
2310 FOR I=1 TO 5
2320 S=S+DDATA(I)
2330 NEXT
2340 FOR I=1 TO 5
2350 DDATA(I)=DDATA(I)/SX100
2360 NEXT
2370 FOR I=0 TO 100 STEP 2
2380 S=1/100*3.142*2
2390 X=INT(SIN(S)*100):Y=INT(COS(S)*100)
2400 IF I=0 THEN LPRINT "M";X;",";Y
2410 LPRINT "D";X;",";Y
2420 NEXT
2430 S=0
2440 FOR I=1 TO 5
2450 S=S+DDATA(I)*3.142*2/100
2460 X=INT(SIN(S)*100):Y=INT(COS(S)*100)
2470 LPRINT"H":LPRINT"D";X;",";Y
2480 NEXT
2490 P=0
2500 FOR I=1 TO 5

```

```

2510 LPRINT "C";C(I)
2520 O=P:P=P+DDATA(I)
2530 OS=O*3.142*2/100:PS=P*3.142*2/100
2540 OX=INT(SIN(OS)*100):OY=INT(COS(OS)*100)
2550 PX=INT(SIN(PS)*100):PY=INT(COS(PS)*100)
2560 START=100:EEND=-100
2570 IF OY>=0 AND PY>=0 THEN EEND=0
2580 IF OY<=0 AND PY<=0 THEN START=0
2590 FOR Y=START TO EEND STEP -ST(I)
2600 J=0
2610 ZA=SQR(10000-Y*Y)
2620 IF Y=0 THEN ZS=3.142/2:GOTO2650
2630 ZS=ATN(ZA/Y)
2640 IF ZS<0 THEN ZS=ZS+3.142
2650 IF OS<ZS AND ZS<PS THEN D(J)=INT(ZA):J=J+1
2660 ZS=3.142*2-ZS
2670 IF OY=0 THEN 2730
2680 X=OX/OY*Y
2690 IF SGN(X)<>SGN(OX) THEN 2730
2700 IF SGN(Y)<>0 AND SGN(OY)<>SGN(Y) THEN 2730
2710 ZT=SQR(X*X+Y*Y)
2720 IF ZT<=100 THEN D(J)=INT(X):J=J+1
2730 IF PY=0 THEN2790
2740 X=PX*Y/PY
2750 IF SGN(X)<>SGN(PX) THEN2790
2760 IF SGN(Y)<>0 AND SGN(PY)<>SGN(Y) THEN2790
2770 ZT=SQR(X*X+Y*Y)
2780 IF ZT<=100 THEN D(J)=INT(X):J=J+1
2790 IF OS<ZS AND ZS<PS THEN D(J)=INT(-ZA):J=J+1
2800 IF Y=0 AND J<>2 THEN D(J)=0:J=J+1
2810 IF J<=2 THEN2890
2820 FOR U=0 TO J-1
2830 M=D(U):MN=L
2840 FOR L=U+1 TO J-1
2850 IF D(L)<M THEN M=D(L):MN=L
2860 NEXT L
2870 D(MN)=D(U):D(U)=M
2880 NEXT U
2890 U=0
2900 IF J<2 THEN2930
2910 LPRINT"M";D(U);",";Y:LPRINT"D";D(U+1);",";Y
2920 U=U+2:IF U<J-1 THEN2910
2930 NEXTY
2940 NEXTI
2950 LPRINT"M130;";(N-1)*15
2960 FOR I=1 TO 5
2970 LPRINT "I"
2980 LPRINT"C";C(I)
2990 LPRINT"J0,20,30,0,0,-20,-30,0"
3000 J=0
3010 LPRINT"M0;";J:LPRINT "J30,0"
3020 J=J+ST(I):IF J<20 THEN 3010
3030 LPRINT"M40,0":LPRINT"P";A*(I)
3040 FOR K=1 TO 11-LEN(A*(I))
3050 LPRINT"P":NEXT
3060 LPRINT"P";DDATA(I);"%"
3070 LPRINT"M0,-30"
3080 NEXT
3090 LPRINT"M0,-100":LPRINT"A"
3100 END


```

4 PEN Colour Graphic  
Micro Plotter/Printer


MCP-40

\*\*\* SIN & COS GRAPH \*\*\*


SIN =  
COS =



\*\*\*\* BAR GRAPH \*\*\*\*



\*\*\*\* PI GRAPH \*\*\*\*



- A INC 40.94 x
- B INC 24.4137 x
- C INC 19.3808 x
- D INC 10.7863 x
- E INC 4.47922 x

## Appendix B

### Graphic Commands Summary

COMMAND	PARAMETER	FUNCTION
LINE TYPE	Lp(P=0 to 15)	Change line type from 0 to 15. p=0 : solid line p=1~15 : dotted(dashed)line
ALL INITIALIZE	A	Reset Pen moves to left margin (without drawing) and returns to Text Mode.
HOME	H	Moves Pen to current origin without drawing line.
INITIALIZE	I	Sets new Origin (i.e., the current Pen location).
DRAW	Dx, y... xn, yn (-999 ≤ x, y ≤ 999)	Draw from current coordinate to specified destination. If there is more than one point the line continues to the second point, etc. Current Origin is used.
RELATIVE DRAW	JΔx, Δy... Δxn, Δyn (-999 ≤ Δx, Δy ≤ 999)	Draws a line from current Pen location X steps to the right and Y steps up.
MOVE	Mx, y (-999 ≤ x, y < 999)	Move(Absolute). Move without drawing to location x steps right (left) and y steps up (down) from the present origin.
RELATIVE MOVE	RΔx, Δy (-999 ≤ Δx, Δy ≤ 999)	Move (Relative). Move without drawing from present location to location x steps to the right (left) and y steps up (down).
COLOUR CHANGE	Cn (n=0 to 3)	Change colour in TEXT mode.
SCALE SET	Sn (n=0 to 63)	Specify the size of the character.
ALPHA ROTATE	Qn (n=0 to 3)	Change print direction. n=0 : left-to-right(normal) n=1 : top-to-bottom n=2 : upside down n=3 : bottom-to-top
PRINT	Pc <sub>1</sub> c <sub>2</sub> ...c <sub>n</sub> (n = ∞)	Print characters in Graphic Mode.
AXIS	Xp,q,r (p=0 or 1) (q = -999 to 999) (r = 1 to 255)	Draw a coordinate axis from present location in direction p=0 : Y-axis p=1 : X-axis q=measurement marks on axis r=interval

# Appendix C

## ASCII Character Set

The following table lists each character the Printer will print and the ASCII code that will print it.

	0	1	2	3	4	5	6	7
0				0	@	P	ˆ	p
1		DC1	1	1	A	Q	a	q
2		DC2	"	2	B	R	b	r
3			#	3	C	S	c	s
4			\$	4	D	T	d	t
5			%	5	E	U	e	u
6			&	6	F	V	f	v
7			'	7	G	W	g	w
8	BS		(	8	H	X	h	x
9			)	9	I	Y	i	y
A	LF		*	:	J	Z	j	z
B	(VT) LU		+	;	K	[	k	{
C			,	<	L	\	l	
D	CR	NC	-	=	M	]	m	}
E			.	>	N	^	n	~
F			/	?	O	_	o	⊠

(NOTE) BS : Back Space  
 LF : Line Feed  
 Lu U : Line Up  
 CR : Carriage return  
 DC1 : Device Control 1 (Text Mode)  
 DC2 : Device Control 2 (Graphic Mode)  
 NC : Next Colour

IF YOU HAVE ANY QUESTION REGARDING THIS PRODUCT,  
PLEASE CONTACT YOUR DEALER.